تحلیل چند مسئله نمونه

مسائلی که جهت نمونه تحلیل شده اند متعلق به کتاب زیر می باشند :

Design of Analog CMOS Integrated Circuits

Author: Behzad Razavi

ISBN: 0-07-238032-2

شماره نمونه ها

Chapter 2	2-5d	2-9a
Chapter 3	3-15a	3-15e
Chapter 4	4-3	4-18
Chapter 5	5-11b	5-16b
Chapter 14	14-3	Figure 14.8

2-9 : Sketch Vx and Ix as a function of time in Fig. 2.46a . The Initial Voltage of C1 is equal to 3V.



α=λ=1 , Vth=.7

For **$V_b - V_{th} < V_x < 3$ Device is in saturation**

Assume $V_b > V_{th}$:

$$I_x = \frac{1}{x}\mu_n C_{ox}\frac{w}{L}\left(V_b - V_{th}\right)^2$$

$$V_x = -\frac{1}{C_1}\int I_x dt + 3^V = 3 - \frac{1}{2}\mu_n C_{ox}\tfrac{w}{L}\left(V_b - V_{th}\right)^2 t$$

Then device goes into triode region , for $0 < V_x < V_b - V_{th}$:

$$I_x = \frac{1}{2}\mu_n C_{ox}\frac{W}{L}\left[2(V_b - 0.7)V_x - {V_x}^2\right] = -\frac{dV_x}{dt}C_1$$

$$\Rightarrow -dt \times \underbrace{\frac{1}{2}\mu_n C_{ox}\frac{W}{L}\times\frac{1}{C_1}}_{\alpha} = \frac{dV_x}{V_x\left[2\left(V_b - 0.7\right) - V_x\right]}$$

$$\Rightarrow -\alpha dt = \left[\frac{1}{V_x} + \frac{1}{2(V_b - 0.7) - V_x}\right]\times\frac{1}{2(V_b - 0.7)}$$

$$\Rightarrow -\alpha(t - t_0) = \left[\ln\frac{V_x}{2(V_b - 0.7) - V_x}\right].\frac{1}{2\left(V_b - 0.7\right)}\Bigg|_{t=t_0, V_x = V_b - 0.7}$$

$$\Rightarrow \frac{2\left(V_b - 0.7\right) - V_x}{V_x} = e^{\psi} \qquad \psi = 2\alpha(V_b - 0.7)\left(t - t_0\right)$$

$$\Rightarrow V_x = \frac{2\left(V_b - 0.7\right)}{1 + e^{\psi}}$$

$$I_x = -c_1\frac{dv_x}{dt} = \frac{4\alpha C_1 (V_b - 0.7)^2 e^{\psi}}{\left(1 + e^{\psi}\right)^2}$$

برنامه 2-9a :

```
.prot
.lib 'logs355V.l' TT
.unprot
.param lmin = .35u
.options list node post
 vb vb 0 1.5

 m vx vb 0 0 nmos w='100*lmin' l='lmin'
 c vx 0 1u


.IC V(vx)=3
.tran  0.1u 1m
.print I(m) V(vx) vbth=PAR('V(vb)-Vth(m)')
.op
.end
```

در شکل زیر ، منحنی بالا نشان دهنده ولتاژ V_x و منحنی پایین نشان دهنده جریان I_d می باشد .



2-5 : Sketch Ix as a function of Vx in Fig. 2.42d as Vx varies from 0 to Vdd.



Vth = 0.7 , α = 0

Drain & Source exchang their roles

V_{GS} = – 0.9 V_{DS} = V_x – 1.9

For V_x < 1.8 :

$$I_x = \frac{1}{2}\mu_p C_{ox}\frac{W}{L}(0.1)^2$$

$$g_m = -\mu_p C_{ox}\frac{W}{L}(0.1)$$

Device remains in the saturation region until V_x = 1.9 – 0.1 = 1.8 ; Then Device goes into triod region.

For 1.8 < V_x < 1.9 :

$$I_x = -\mu_p C_{ox}\frac{W}{L}\left[(-0.1)(V_x-1.9)-\frac{1}{2}(V_x-1.9)^2\right]$$

$$g_m = \mu_p C_{ox}\frac{W}{L}(V_x-1.9)$$

For V_x > 1.9 : Source and Drain exchange their roles again when V_x = 1.9

For V_x >1.9 , Device operates in the triode region
V_{GS} = 1 – V_x , V_{DS} = 1.9 –V_x

$$I_x = \mu_p C_{ox}\frac{W}{L}\left[(1.8-V_x)(1.9-V_x)-\frac{1}{2}(1.9-V_x)^2\right]$$

$$g_m = -\mu_p C_{ox}\frac{W}{L}(1.9-V_x)$$

For 0 < V_x < 1.8 :

$$I_x = -\frac{1}{2}\mu_p C_{ox}\frac{W}{L}(0.1)^2$$

$$g_m = -\mu_p C_{ox}\frac{W}{L}(0.1)$$

For 1.8 < V_x < 3 :

$$I_x = +\mu_p C_{ox}\frac{W}{L}(0.1)$$

$$g_m = \mu_p C_{ox}\frac{W}{L}(V_x-1.9)$$

برنامه 2-5d :

```
.prot
.lib 'logs355V.l' TT
.unprot
.param lmin = .35u

 vb1 vb1   0     1.9
 vb2 vb2   0     1
 vx  vx    0

 m vb1 vb2 vx vx pmos w='10*lmin' l='lmin'

.DC vx 0 3 0.01
.probe I(m) V(vx)
.print Ix=PAR('I(vx)*-1') vgth=PAR('V(vb1)-vth(m)')
.end
```

شکل های زیر از پنجره Avanwaves بدست آمده اند :





3-15 : Sketch Vout versus Vin for circuit of Fig. 3.67a as Vin varies from 0 to VDD. Identify important transition points.



R_D = 0.5k , VDD = 5V
R_F = 20k , V_b = 1V

V_{in} = 0
V_{GS} > V_{th} → 1-V_{in} > 0.7 → 1 > 0.7 So M_1 is ON

Assume M_1 is in the saturation region

V_{out} = V_D > 0.3 → M_1 is in the saturation region

$$I_D = \frac{1}{2}\mu_n C_{ox}\frac{W}{L}(1-0.7)^2 \longrightarrow \frac{V_o}{20}+\frac{V_o-5}{0.5}+0.603=0 \longrightarrow V_o = 4.58$$

For V_{in} < 0.3 M_1 is ON . For greater values of V_{in} M_1 is off , and the output voltage remains linear as V_{in} arises.

برنامه :

```
.lib 'logs355V.l' TT
.param lmin = '.35u'

 vdd        vdd   0      5
 vb   vb    0     1
 vin vin    0

 m1 out vb vin 0 nmos w='20*lmin' l=lmin
 rd vdd out 500
 rf out vin 20k

.DC vin 0 5 0.001
.end
```



3-15 : Sketch Vout versus Vin for circuit of Fig. 3.67e as Vin varies from 0 to VDD. Identify important transition points.



R_F = 10k , R_s = 1k
VDD = 3V
0 < V_i < 3

For V_i = 0 , M_1 is OFF , so V_{out} = 0
For $V_i - V_{out}$ > 0.7 ; M_1 turns ON

V_{out} = 1/11 V_{in} → 10/11 V_{in} > 0.7 →
So For V_{in} > 0.77 M_1 turns ON and meanwhile V_{out} = 1/11 V_{in}
So M_1 is in the saturation region

Saturate Condition :

3 > V_{in} – 0.7 → V_{in} < 3.7

برنامه :

```
.lib 'logs355V.l' TT
.param lmin = '.35u'
 vdd vdd 0 3
 vin vin 0

 m1 vdd vin out 0 nmos w='100*lmin' l=lmin
 rf vin out 10k
 rs out 0 1K

.DC vin 0 3 0.001
.end
```



4-3 : Construct the plots for Fig. 4.8(c) for a differential pair using PMOS transistors.



R_D = 10k
V_b = 1V
V_{DD} = 3.3V , 0 < $V_{in.CM}$ < 3







برنامه :

```
.lib 'logs355V.l' TT
.param lmin = .35u
.options list node post
 vb  vb    0 1
 vdd vdd   0 3.3
 vcm vcm   0

 m1 x vcm p vdd pmos w='100*lmin' l=lmin
 m2 y vcm p vdd pmos w='100*lmin' l=lmin
 m3 p vb vdd vdd pmos w='100*lmin' l=lmin
 rd1 x 0 10k
 rd2 y 0 10k

.DC  vcm 0 3.3 .001
.print v(p) I(rd1) I(rd2) v(x) v(y)
.end
```

شکل موج گره X و Y که کاملاً مانند هم می باشند :



شکل موج جریان مقاومتهای RD که برابر هم می باشند :



شکل موج ولتاژ گره P :



4-18 : Assuming all of the transistors in the circuit of Fig. 4.39c are saturated and $\lambda \neq 0$, calculate the small-signal differential voltage gain of circuit.



VDD = 3V
Vb = 0.75V
Vcm = 1.5V , DC
Vdm = 10mV , 1KHz , AC
M1 , M2 : W/L = 40
M3 , M4 : W/L = 8
M5 : W/L = 200

If we neglect ro3 & ro4 at the moment , we have :

$$1: I_x = g_{m_3} V_2 \quad g_{m_3} = g_{m_4} = g_{m_{3,4}}$$

$$2: I_x = -g_{m_4} V_1$$

$$1+2: 2Ix = -g_{m_{3,4}}(V_1 - V_2) = -g_{m_{3,4}} V_x$$

$$\Rightarrow \frac{V_x}{I_x} = -\frac{2}{g_{m_{3,4}}}$$



$$A_v = -g_{m_1}\left(r_{o_1} \| r_{o_3} \| -1/g_{m_3}\right)$$

$$A_v = -\frac{g_{m_1}}{1/r_{o_1} + 1/r_{o_3} + g_{m_3}} \qquad \left(1/r_{o_1} + 1/r_{o_3} \succ g_{m_3}\right)$$



if $g_{m_3} \geq 1/r_{o_1} + 1/r_{o_3}$ then the circuit is not stable and small-signal model is not valid.

برنامه :

```
.prot
.lib 'logs355V.l' TT
.unprot
.param lmin = .35u
.options list    node  post

 vdd vdd 0 3
 vin1 vin1 0 sin(1.5  10m   1k  0 0 0)
 vin2 vin2 0 sin(1.5  10m   1k  0 0 180)
 vb vb 0 .75

 m1  out1  vin1   p    0   nmos w='40*lmin'  l='lmin'
 m2  out2  vin2   p    0   nmos w='40*lmin'  l='lmin'
 m3  out1  out2   vdd  vdd pmos w='8*lmin'   l='lmin'
 m4  out2  out1   vdd  vdd pmos w='8*lmin'   l='lmin'
 m5  p     vb     0    0   nmos w='400*lmin' l='2*lmin'

.tran .01m 10m
.op
.print I(m1) I(m2) gain = PAR('(V(out1)-V(out2))/(V(vin1)-V(vin2))')
.end
```

Time	Current M1	Current M2	Gain	Time	Current M1	Current M2	Gain
1.00000m	227.8960u	227.8960u	7.8409	11.00000m	227.8960u	227.8960u	7.8409
2.00000m	227.8960u	227.8960u	7.8409	12.00000m	227.8960u	227.8960u	7.8409
3.00000m	227.8960u	227.8960u	7.8409	13.00000m	227.8960u	227.8960u	7.8409
4.00000m	227.8960u	227.8960u	7.8409	14.00000m	227.8960u	227.8960u	7.8409
5.00000m	227.8960u	227.8960u	7.8409	15.00000m	227.8960u	227.8960u	7.8409
6.00000m	227.8960u	227.8960u	7.8409	16.00000m	227.8960u	227.8960u	7.8409
7.00000m	227.8960u	227.8960u	7.8409	17.00000m	227.8960u	227.8960u	7.8409
8.00000m	227.8960u	227.8960u	7.8409	18.00000m	227.8960u	227.8960u	7.8409
9.00000m	227.8960u	227.8960u	7.8409	19.00000m	227.8960u	227.8960u	7.8409
10.00000m	227.8960u	227.8960u	7.8409	20.00000m	227.8960u	227.8960u	7.8409

element	0:m1	0:m2	0:m3	0:m4	0:m5
model	0:nmos.3	0:nmos.3	0:pmos.3	0:pmos.3	0:nmos.12
region	Saturati	Saturati	Saturati	Saturati	Saturati
id	227.8958u	227.8958u	-227.8958u	-227.8958u	455.7917u
ibs	-607.1540a	-607.1540a	2.221e-19	2.221e-19	-9.668e-19
ibd	-607.9433a	-607.9433a	27.5661f	27.5661f	-5.7439f
vgs	970.5123m	970.5123m	-1.9751	-1.9751	750.0000m
vds	495.3721m	495.3721m	-1.9751	-1.9751	529.4877m
vbs	-529.4877m	-529.4877m	0.	0.	0.
vth	729.7303m	729.7303m	-659.7967m	-659.7967m	606.3598m
vdsat	217.9774m	217.9774m	-1.0486	-1.0486	155.8455m
beta	8.7083m	8.7083m	304.7593u	304.7593u	40.4436m
gam eff	596.8141m	596.8141m	461.6218m	461.6218m	485.3974m
gm	1.5784m	1.5784m	271.8181u	271.8181u	5.0525m
gds	50.1484u	50.1484u	20.2977u	20.2977u	37.1273u
gmb	341.2814u	341.2814u	52.0776u	52.0776u	1.4758m
cdtot	18.0613f	18.0613f	3.7510f	3.7510f	188.2502f
cgtot	22.0237f	22.0237f	3.9351f	3.9351f	369.9977f
cstot	35.7382f	35.7382f	9.0772f	9.0772f	554.8506f
cbtot	34.0622f	34.0622f	9.2340f	9.2340f	434.6594f
cgs	16.7404f	16.7404f	3.0854f	3.0854f	293.9998f
cgd	3.9257f	3.9257f	657.6318a	657.6318a	39.0808f

5-11 : For circuit in Fig. 5.35b, Sketch Vx and Vy as a function of V1 for 0<V1<VDD. Assume the transistor in circuit are identical.



R_1 = 4.8k , R_2 = 1.8k
Iref = .45mA , VDD = 3.3V
0 < V_1 < 3.3

For Better Visualization :
0 < V_1 < 2

When V_1 = 0 , $I_1 = I_2$ = Iref and $V_x = V_y$, $\Delta I = 0$

As V_1 increases , I_2 gradually decreases and part of Iref flows throw R_2 , V_x increases and $V_y = V_x - R_2 * \Delta I$, decreases .

Finally when V_1 is large enough such that M_2 turns off
$V_y = V_x - R_2$.Iref and both V_x and V_y are set at a constant voltage.

برنامه :

```
.lib 'logs355V.l' TT
.param lmin = .35u
.options list    node  post

 vdd vdd 0 3.3
 v1 v1 0

 m1 vy vx 0 0 nmos w='20*lmin'
 +l=lmin
 m2 vx vx v1 0 nmos w='20*lmin'
 +l=lmin
                    →
```

```
 r1 vdd vy 4.8k
 r2 vx vy 1.8K
 Iref vdd vx .45mA

.DC v1 0 2 0.01
.Probe V(vx) V(vy) I(m1) I(m2)
.op
.end
```



5-16 : Sketch Vx and Vy as a function of time for circuit Fig 5.40b . Assume the transistors in circuit are identical.



R_1 = 47k , VDD = 3V
C_1 = 10pF , Iref = 300uA

M_2 is ON with fixed $V_x = V_{GS}$, C_1 is charged with I_1 until M_1 turns OFF.

$V_y = VDD - I_1R_1$ Where I_1 goes from Iref to zero

برنامه :

```
.lib 'logs355V.l' TT
.param lmin = .35u
.options list node post

 vdd   vdd 0 3
 Iref  vdd x 300u

 m1 y x d 0 nmos w='100*lmin' l=lmin
 m2 x x 0 0 nmos w='100*lmin' l=lmin
 r1 vdd y 47K
 c1 d 0 10p ic=0

.tran .1n 500n UIC
.Print V(x) V(y) I(r1) V(d) vgs1=PAR('v(x)-v(d)')
.op
.end
```



14-3 : For the circuit of Fig. 14.12 , determine the minimum value of ISS that guarantees oscillation.



Each stage must provide a small-signal gain of 2.
That is , $g_{m1}R_1 = 2$.
With small swings, each transistor carries a half of the tail current
For squre-law devices , therefor , we have :

$$g_{m1}R_1 = 2 = \sqrt{\mu_n C_{ox}\frac{W}{L} I_{ss}} R_1 = 2$$

$$I_{ss} \geq \frac{4}{\mu_n C_{ox}\frac{W}{L} R_1^{\ 2}}$$

برنامه :

```
.prot
.lib 'logs355V.l' TT
.unprot

.param lmin = .35u
.param alpha = 100
.param Rvalue = 12k
.param Issv = .15m

.options list node post

 vdd vdd 0 3.3

 m1 x1 x3 p12 0 nmos w='alpha*lmin' l=lmin
 m2 y1 y3 p12 0 nmos w='alpha*lmin' l=lmin
 m3 x2 y1 p34 0 nmos w='alpha*lmin' l=lmin
 m4 y2 x1 p34 0 nmos w='alpha*lmin' l=lmin
 m5 x3 y2 p56 0 nmos w='alpha*lmin' l=lmin
 m6 y3 x2 p56 0 nmos w='alpha*lmin' l=lmin

 r1 vdd x1 Rvalue
 r2 vdd y1 Rvalue
 r3 vdd x2 Rvalue
```

ادامه برنامه :

```
r4 vdd y2 Rvalue
r5 vdd x3 Rvalue
r6 vdd y3 Rvalue

Iss12   p12        0       Issv
Iss34   p34      0         Issv
Iss56   p56        0       Issv

.ic V(x1)=1;
.tran .001n 10n
.Probe ac V(x1)
.op
.end
```

شکل موج خروجی ازترمینال درین ترانزیستورها و در پایین درین یکی از آنها :





Fig. 14.8 : Set the value for each component in order to have an oscillation .



برنامه :

```
.prot
.lib 'logs355V.l' TT
.unprot

.param lmin = .35u
.param alpha = 120
.param RD = 680
.param CL = 120p

.options list    node  post
 vdd vdd 0 3

 m1 E G 0 0 nmos w='alpha*lmin'
  +l=lmin
 m2 F E 0 0 nmos w='alpha*lmin'
 +l=lmin
 m3 G F 0 0 nmos w='alpha*lmin'
 +l=lmin
              →

 r1 vdd E RD
 r2 vdd F RD
 r3 vdd G RD

 c1 E 0 CL
 c2 F 0 CL
 c3 G 0 CL

.ic V(E)=1;
.tran .01n 2u
.Probe V(G)
.op
.end
```